
News 
Signal version 3.07 released 

The latest version of Signal is now available to download from our website by following 
the links below. This version now includes new support for voltage and patch clamp 
experiments. See the Signal feature in this newsletter for part 1 of our series on clamp 
sampling. 

 

Local training sessions 
In addition to our annual UK and US training days, we are also happy to offer local 
training sessions for Spike2 and Signal at universities and institutions. These smaller 
training sessions can be tailored towards particular requirements for research groups or 
departments. If you are interested in organizing a local training seminar for your group or 
department, please contact sales@ced.co.uk.   

 Latest versions of Spike2 and Signal 
The latest updates for Spike2 and Signal are available from the CED downloads page, or 
by clicking on the links in the table below. Demonstration versions of the latest software 
are also available. 
 

 Spike2 downloads Signal downloads 
Spike2 version 5.14  Signal version 3.07 

 Spike2 version 4.24  Signal version 2.16 

 Spike2 demo  Signal demo 

A full list of the new features and changes in the latest software versions is available from 
the website. 

 

See us at… 
5th Polish Electrophysiology Forum 
Lodz 
Poland 
June 7th - 9th  

The Physiological Society AGM 
University College London 
July 5th  - 7th  

5th Forum of European Neuroscience (FENS) 
Austria Centre 
Vienna 
July 8th  - 12th  

Spike2   

S1-S2 heart pacing protocol 
Q. I am hoping to start recording cardiac action potentials using the S1-S2 pacing 

protocol as a stimulus and would like to use the 1401 to generate and control the 
pulse train. Is this possible using the output sequencer in Spike2? 

A. The Spike2 sequencer can be used to create a variety of stimulus protocols to 
use when sampling. The S1-S2 protocol is widely used as a pacing stimulus for 
recording cardiac action potentials, where the first stimulus (S1) is used to pace 
the heart at a fixed cycle length. The S2 stimulus is then applied at an initial 
interval after the S1 pulse. The time between the S1 and S2 stimulus is 
progressively reduced until the stimulus either a) fails to illicit a response; or b) 
generates a rapid atrial response. The object is usually to determine the shortest 
effective refractory period that still results in the generation of the cardiac action 
potential. 

  

        
 

          
 Basic S1-S2 protocol. The S2 interval reduces by a fixed amount 

for each stimulus cycle  

 

The attached script S1S2protocol.s2s generates a script toolbar which 
allows the user to open a new data file for sampling, set the parameters of the 
stimulation and apply and stop the stimulus when sampling. The attached 
sequence file S1S2protocol.pls must be loaded before running the script. 
The stimulus properties set in the script toolbar are copied to the sequencer 
when they are updated during sampling. Using a text sequence file gives greater 
flexibility and allows sequencer variables to be updated and modified from the 
script. For this example of implementing a basic S1-S2 protocol, the user can 
change the S1 cycle length, the initial interval between the S1-S2 stimulus and 
the reduction in S1-S2 time per cycle during the experiment. The sequencer 
generates the S1 pulses from digital output 0 and the S2 pulses from digital 
output 1 of the 1401.   

  

Signal 

 

Clamp sampling: part 1  

Q. I am using Signal for voltage clamp experiments and would like to know how to 
make best use of the new clamp support features. 

A. The latest version of Signal includes extra sampling support for whole-cell and 
single-channel clamping experiments. When Signal is run for the first time, it will 
give the option of including clamp support. The Edit preferences dialog can be 
used to enable or disable this option at any other time. The features enabled by 
the clamp option are the new clamp sampling support and the existing leak 
subtraction analysis and single-channel idealized trace functions. These features 

'$S1S2protocol|Controls sampling from a toolbar and allows the user to specify cycle length, intervals
'and other parameters for an S1-S2 protocol. The sequencer file S1S2protocol.pls should be loaded prior 
'to running the script.

Var count%:=8;          'Number of pulses
Var S1Int%:=125;        'Pulse interval
Var IntTDel%:=1000;     'S1-S2 delay
Var Reduc%:=10;         'Interval reduction
Var DelayT%:=10000;     'Delay between cycles
Var ok%;
Var lastN%, Num%;
Var go%;
Var data%;
var gFloat%[20];              'global for floating window states
LastN%:=Num%;


New%();
HideAll();
DoToolbar();   'Try it out
RestoreAll();
Halt;

Func DoToolbar()    'Set your own name...
ToolbarClear();     'Remove any old buttons
ToolbarSet(0,"",Idle%);
ToolbarSet(1, "&Quit");    'This button returns its number
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(4,"&New file", New%);
ToolbarSet(6, "Inter&vals", Intervals%);   'Link to function
ToolbarSet(7, "& Single", Single%);   'Link to function
ToolbarSet(8, "&Cycle", Cycle%);   'Link to function
ToolbarSet(9, "&Stop Pulses", StopPulses%);   'Link to function
If go%<1 then
     Toolbarenable(3,0);
     ToolbarEnable(7,0);
	 ToolbarEnable(8,0);
	 ToolbarEnable(9,0);
Endif;
return Toolbar("Waiting", 1023);
end;

func New%()								'New sampling window
View(LogHandle());					'Make log view the current view
EditSelectAll();						'Select all text in log view
EditClear();							'Delete it
Window(0,80,100,100);				'Display it at the bottom of the screen
WindowVisible(1);						'Make it visible

if data%>0 then						'If there is already a data view open...
  View(data%);							'...close it
  FileClose();                   'No point setting data% to 0...
endif;

data%:=FileNew(0,1);					'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file");Halt() endif;
Window(0,0,100,100);					'Make data window in top bit of screen
XRange(0,10);

ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(2,1);					'Disable "Sample stop" button
ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

func Start%()							'If "Start" is pressed
SampleStart();							'Start sampling
ToolbarEnable(4,0);					'Disable "New file" button
ToolbarEnable(3,1);					'Enable "Sample stop" button
ToolbarEnable(2,0);					'Disable "Sample start" button
ToolbarEnable(1,0);					'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;								'Stay with toolbar
end;

func Stop%()							'If "Stop" is pressed
SampleStop();							'Stop sampling
if (data%>0) and                 'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);					'Enable "New file" button
ToolbarEnable(3,0);					'Disable "Sample stop" button
ToolbarEnable(1,1);					'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;								'Stay in toolbar
end;

Func Idle%();

Var String$;

If SampleStatus()=2 then        'Check that sampling is in progress
    Num%:=SampleseqVar(4);      'Get next stimulus
    Num%:=num%+4;
    If num%<> lastN% then
        
        String$:=Print$("Current S2 interval = %d ms", Num%);   'Print the current S2 interval to the toolbar
        ToolbarText(String$);
        Sampletext(String$);
        LastN%:=num%        
    Endif
endif
Return 1
End

Func Intervals%()   'Parameter settings
DlgCreate("Stimulus parameters");  'Start new dialog
DlgInteger(1,"Number of pulses per train",1,100,0,0,1);
DlgInteger(2,"S1 interval (ms)",5,1000,0,0,5);
DlgInteger(3,"S2 Interval - Initial (ms)",5,1000,0,0,5);
DlgInteger(4,"S2 reduction (ms)",0,500,0,0,5);
DlgInteger(5,"Delay between trials (ms)",0,10000,0,0,100);

DlgButton(0,"Cancel");
DlgButton(1,"OK");
ok% := DlgShow(Count%,S1Int%,IntTDel%,Reduc%,DelayT%);    'ok is 0 if user cancels,  variables updated if not

If ok%=1 then
    sampleseqvar(2,count%);         'Copy the values to the sequencer
    sampleseqvar(3,(S1Int%-4));
    sampleseqvar(4,(IntTDel%-4));
    sampleseqvar(5,Reduc%);
    SampleSeqVar(6,(DelayT%-14));
    Go%:=1;
else
    Message("Parameters not updated");  'Or warn user if Cancel is pressed
endif;


Toolbarenable(7,1);
Toolbarenable(8,1);
return 1;  'This leaves toolbar active
end;

Func Single%()   'Single S1-S2 stimulus
Samplekey("P");
Toolbarenable(7,1);
Toolbarenable(8,0);
Go%:=1;
return 1;  'This leaves toolbar active
end;

Func Cycle%()   'Button 5 routine
Samplekey("G");
Toolbarenable(7,0);
Toolbarenable(8,0);
'Go%:=1;
return 1;  'This leaves toolbar active
end;

Func StopPulses%()   'Button 5 routine
Samplekey("S");
'Toolbarenable(4,1);
'Toolbarenable(5,1);
Go%:=0;
return 1;  'This leaves toolbar active
end;

proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

have all been included in the clamp options so that non-clamping users can 
disable them when they are not needed.  

  

Clamp setup 

 The following guidelines assume a basic sampling configuration that uses the 
first two available ADC ports of the 1401, with a one second sweep length and a 
sample rate of 10kHz. The attached configuration ClampExample.sgc is set up 
with the examples detailed below and can be used as a starting point for carrying 
out clamping experiments with the latest version of Signal. 

With the clamping options selected, the sampling configuration has an additional 
Clamp tab where you can define up to two simultaneous clamp setups. This 
allows easy set-up of the channels recording the stimulus and response from the 
cell or patch and the DAC output used to control the stimulus for the selected 
experiment type. Once Signal has this information it can automatically scale the 
data and perform further clamp specific analysis. The State for resistance 
measurements can be used in Multiple states mode to nominate a particular 
State (set of pulse outputs) to monitor the membrane resistance on-line. In this 
simple example we are only going to specify a single set of outputs. We will set 
the State field to 0, which defaults to the current set of outputs for on-line 
measurements. How to use Multiple states for resistance measurements will be 
covered in more detail in the next part of our clamp sampling series. 

 
Example clamp setup  

  

Channel setup and units 

For voltage-clamp experiments the applied stimulus is voltage and the response 
is current and vice versa for current-clamp setups. The channels used to record 
the stimulus and response data must be set to use valid units in the Port setup 
tab of the sampling configuration. The units for a voltage channel must include 
the character ‘V’ (for example, mV) and the units for a current channel must 
include the character ‘A’ (or example, pA), so that on-line measurements give 
sensible values. The control DAC should also be set to match the units of the 
stimulus channel for the selected experiment type (see Stimulus setup below). 
Incorrect units for any aspect of the clamp setup will cause the text to be 
displayed in red and Signal will generate an error message when ready to 
sample. 

Port setup parameters for the stimulus channel 

             

 Stimulus setup 

The units for the control DAC are set in the Outputs tab of the sampling 
configuration. The stimulus pulse is set using the graphical pulse configuration 
editor, which is available from the Configure pulses button. You can also access 
the pulse configuration during sampling to make adjustments and modifications 
on-line. Supported outputs for clamp experiments are constant or varying 
amplitude square pulses and square pulse trains. If more than one pulse is used 
for a given state you can define the one to use for resistance measurements (RM) 
by setting the pulse ID in the dialog, as below. 

  
 

 

 

 

 

 

 

Outputs tab and pulse configuration with the second pulse specified for 
measurements 

Ready to sample… 

Following the setup above you should now be ready to sample. The sampling 
configuration is checked for the correct settings for clamp experiments when a 
new data file is opened. Signal will generate an error message to alert the user if 
any incompatibilities are found. 

In the next eNewsletter - Part 2: Running a clamp experiment with on-line control 
and analysis. 

Scripts: Spike2  
Q. I have been making use of the Rerun feature in Spike2 for presentations, to show 

the data occurring in ‘real-time’. It would be really useful if I could somehow get 
sound output of a data channel as well during a rerun. Is there any way to do this?  

A. The attached script Rerun with sound.s2s can be used to rerun a data file 
and nominate a channel to be played out from the sound card during the replay. 
The script generates a toolbar from which the user can open data files, set the 
channel to play and run/stop the file.  

Scripts: Signal  
Q. I am recording from interneurones using Signal and would like get information 

about the spiking data. Is there a way to get the number and timing of 
intracellular spikes for each frame?  

A. The attached script ThreshCount.sgs counts spikes in a Signal data file that 
exceed a user-set threshold. The number of spikes per frame and the time at 
which they occur within the frame is printed to the Log view  

Did you know…? 
We also manufacture amplifiers for life science research. The 1902 amplifier can be 
specified for general purpose transducer use or with medical-specification isolation 
suitable for EMG, ECG, EEG and evoked response measurements as well as many other 
research applications. 1902’s are available as single, dual and quad-channel units, and 
multiple units can be connected together up to a maximum of 32 channels. They are 
controlled directly from Spike2, Signal or dedicated calibrate and control software via the 
host computers RS232 serial port, giving access to the gain and filter settings for each 
channel. The 1902 also has options for more specialised applications, including magnetic 
stimulus artefact clamps, skin conductance modules and input buffer options for multi-
channel common-reference studies. 

Recent questions  
Q. I am analysing large files that take a long time to display when I open them in 

Spike2. Is there a way to stop them drawing all of the data in the file? 

A. You can break out of long drawing or calculation operations in Spike2 using the 
Ctrl+Break keyboard shortcut. This is useful if you have inadvertently 
displayed or analysed a very large section of data. There is also an option in the 
General tab of the preferences dialog to ignore resource file X range settings to 
give a fast initial display by only showing the first second of data when a large file 
opens. 

User Group   
 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

'$Rerun with sound.s2s|Spike2 script to open a data file and rerun with one channel 
'output simultaneously through the computer sound card 


var data%;      'handle for the data file
var tScale := 1;    'Output speed for the playing channel
var chan%;      'The channel to play
var ok%;        'Checks for OK 
var setFlag% := 0;  'Flag to show channel selected for play
var rerunStatus%;   'Returns the file status


ToolbarSet(1, "Quit");      'This quits the script
ToolbarSet(3, "Open file", Open%);      'Links to Open file functions
ToolbarSet(4, "Settings",GetSettings%); 'Links to Settings
ToolbarSet(5, "Run", DoRerun%);         'Links to rerun
ToolbarEnable(4,0);         'Disable Settings button
ToolbarEnable(5,0);         'Disable ReRun button
Toolbar("Data rerun",1023);



Func Open%()
if data% > 0 then       'If we already have a data file open
	View(data%);        'view it..
	FileClose(0,-1);    '..and close it
endif;
data% := FileOpen("",0,3);  'Prompt user for data file to use
if data% then               'If data file opened successfully
	ToolbarEnable(4,1);     'Enable the Settings button
endif;
return 1;
end;


Func GetSettings%()
View(data%);        'Make sure data is the current view
DlgCreate("Rerun setup");   
DlgChan(1,"Channel to output",1);
DlgReal(2, "output speed", 0.25,4);     'Settings for playing the channel
ok% := DlgShow(chan%,tScale);
if ok% then
	setFlag% := 1;          'If settings OK'd
	ToolbarEnable(5,1);     '..Enable the Rerun button
endif;
return 1;
end;


Func DoRerun%()
View(data%);
Rerun(1,0,MaxTime(),tScale);    'Start file rerun at chosen speed
PlayOffline(chan%,-1,0,MaxTime(),1,tScale);     'Start playing the channel through the soundcard
ToolbarSet(5, "Stop", Stop%);   'Update toolbar
ToolbarSet(0, "Idle", RerunMonitor%);   'Set Idle routine 
ToolbarEnable(3,0);     'Disable Open file button
ToolbarEnable(4,0);     'Disable Settings button
return 1;
end;

Func Stop%()
View(data%);    'Make sure data is current
Rerun(0);       'Stop rerun
PlayOffline(-1);    'Stop playing of channel
ToolbarEnable(3,1); 'Enable Open file button
ToolbarEnable(4,1); 'Enable Settings button
ToolbarSet(5, "Run", DoRerun%); 'Reset Run button
return 1;
end;


Func RerunMonitor%()
View(data%);
rerunStatus% := Rerun();    'Check for running status
if rerunStatus% = 0 then    ' if status is stopped
	ToolbarClear(0);        'Clear idle routine
	ToolbarEnable(3,1);     'Enable Open file button
	ToolbarEnable(4,1);     'Enable Settings button
	ToolbarSet(5, "Run", DoRerun%);     'Reset Run button
endif;
return 1;
end;

'$ThreshCount|Script to count intracellular spikes that exceed a threshold. Number and times of spikes
'for each frame are printed to the Log view and displayed

var count%;
var last;
var current;
var gotSpikeFlag% := 0;
var frameCounter%;
var data%;
var spikeTimes[1000];		'Array to hold actual times of spikes
var printCount%;

data% := FrontView();		'Get current view
if ViewKind(data%) <> 0 then		'If this is not a data file
	Message("Current view is not a time view\nPlease open a time view");
	data%:= FileOpen("",0,1);		'Prompt user to open a file
	endif;
if data% < 0 then		'If no luck
	Message("Unable to open a data file for analysis");		'Warn user
	halt;			'and stop script
endif;

View(LogHandle());		'Get the Log view
EditSelectAll();	'Select all text 
EditClear();		'and clear
WindowVisible(0);
Window(20,20,80,80);
PrintLog("Frame\tSpikes\n");	'Print titles
View(data%);
HCursorDelete(-1);		'Delete existing horizontal cursors
HCursorNew(1);				'Set horizontal cursor on channel 1
Interact("Set horizontal cursor at threshold level and press OK",1023); 'Allow user to set level


for frameCounter% := 1 to FrameCount() do			'Loop through each frame
	count% := 0;											'Reset spike counter
	Frame(frameCounter%);								'Go to frame	
	current := MinTime();								'Set variable to beginning of frame
	CursorSet(1,MinTime());								'Position cursor at beginning of frame 	
	CursorMode(1,7);										'Set cursor to loook for rising threshold crossings	
	repeat													'Loop through each crossing	
		gotSpikeFlag% := 0;								'Set flag to zero	
		last := Current;									'Reset variable
		CursorActiveSet(1,1,current,MaxTime(),HCursor(1));		'Make cursor find next threshold crossing
		current := Cursor(1);							'Get cursor position			
		if current > last then							'If cursor has moved (spike detected)
			spikeTimes[count%] := Cursor(1);
			count% += 1;									'Add 1 to spike count
			last := current;								'Update variable
			gotSpikeFlag% := 1;							'Flag  to say spike has been found
		endif;													
	until gotSpikeFlag% = 0;							'Stop when no spike found		
	if count% > 0 then
		ArrMul(spikeTimes[],1000);						'Convert times to ms	
		PrintLog("%d\t%d",frameCounter%,count%);	'Print frame number and spike count to log window
		for printCount% := 0 to count%-1 do
			PrintLog("\t%g",spikeTimes[printCount%]);
		next;
		PrintLog("\n");
	else
		PrintLog("%d\t%d\t\n",frameCounter%,count%);	'Print frame number and spike count to log window		
	endif;
next;

FrontView(LogHandle());		'Display Log view

